unittest
{
	assert( [ "3 3", "1 3 4", "1 3 6" ].parse.expand.solve == 60 );
	assert( [ "6 5", "-790013317 -192321079 95834122 418379342 586260100 802780784", "-253230108 193944314 363756450 712662868 735867677" ].parse.expand.solve == 835067060 );
}



import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	input.popFront;
	auto xs = input.front.split.to!( long[] );
	input.popFront;
	auto ys = input.front.split.to!( long[] );
	return tuple( xs, ys );
}

auto solve( long[] xs, long[] ys )
{
	// すべての長方形の横辺の長さの合計を計算
	auto xt = 0L;
	for( auto i = 1L; i < xs.length.signed; i++ )
	{
		xt = xt.addMod( xs[ i ].subMod( xs[ i - 1 ] ).mulMod( i ).mulMod( xs.length.signed - i ) );
	}
	
	// すべての長方形の縦辺の長さの合計を計算
	auto yt = 0L;
	for( auto i = 1L; i < ys.length.signed; i++ )
	{
		yt = yt.addMod( ys[ i ].subMod( ys[ i - 1 ] ).mulMod( i ).mulMod( ys.length.signed - i ) );
	}
	
	// 横辺 * 縦辺
	return xt.mulMod( yt );
}

// 足し算
auto addMod( long MOD = 10L ^^ 9 + 7 )( long x, long y )
{
	return ( x + y ) % MOD;
}

// 引き算
auto subMod( long MOD = 10L ^^ 9 + 7 )( long x, long y )
{
	return ( x + MOD - y ) % MOD;
}

// 掛け算
auto mulMod( long MOD = 10L ^^ 9 + 7 )( long x, long y )
{
	return x * y % MOD;
}
